
FORM & Compton scattering. (November 16, 2019)

A. Context

Figure (1) gives the Feynman diagrams corresponding to the two lowest-order processes contribut-

ing to Compton scattering. As per Griffiths’ (2008, Example 7.4) the probability amplitude M1

associated with the first of those Feynman diagrams is:

where ū4 is the (adjoint) spinor associated with the outgoing electron, u1 is the spinor representing

the incoming electron, ǫµ(2) are the components of the polarization 4-vector for the incoming photon

(etc.), and the ‘slash’ notation ✁q ≡ qλγλ. Spin and momentum labels have been dropped, i.e. more

completely, for example, ū4 = ū(s4)(p4) with two possible spin states (s4 = 1, 2). (Implicitly, the

mc factor is multiplied by the identity matrix.)

Figure 1: Feynman diagram of the two lowest-order (i.e. 2-vertex) processes contributing to
Compton scattering (Griffths 2008, p246). Associated with the diagram on the left is an amplitude
M1.

Integration sends q → (p1−p3), and we drop the factor (2π)4 δ4(p1+p2−p3−p4). Then noting

that for fixed index µ the quantity ǫµ(2) is just a scalar (and ditto ǫ∗κ(3)) so that ǫµ(2)ū4 = ū4 ǫµ(2)

(etc.), the amplitude associated with diagram 1 can be written

M1 =
g2e

(p1 − p3)2 −m2c2
[ū4 ✁ǫ2 (✁p1 − ✁p3 +mc) u1 ✁ǫ

∗

3 ] , (2)

where ✁ǫ2 ≡ ǫµ(2)γ
µ. Analogous steps for the second diagram give amplitude

M2 =
g2e

(p1 + p2)2 −m2c2
[ū4 ✁ǫ

∗

3 (✁p1 + ✁p2 +mc) u1 ✁ǫ2 ] . (3)

To evaluate the cross section for Compton scattering the quantity

|M|2 = M1M
∗

1 +M2M
∗

2 +M1M
∗

2 +M2M
∗

1 (4)

is needed. After some manipulation one obtains

M1M
∗

1 = A2 ū4 ✁ǫ2 [✁p1 − ✁p3 + mc] ✁ǫ
∗

3 u1 ū1 ✁ǫ3 [✁p1 − ✁p3 + mc] ✁ǫ
∗

2 u4 , (5)

M2M
∗

2 = B2 ū4 ✁ǫ
∗

3 [✁p1 + ✁p2 + mc] ✁ǫ2 u1 ū1 ✁ǫ
∗

2 [✁p1 + ✁p2 + mc] ✁ǫ3 u4 , (6)

M1M
∗

2 = AB ū4 ✁ǫ2 [✁p1 − ✁p3 + mc] ✁ǫ
∗

3 u1 ū1 ✁ǫ
∗

2 [✁p1 + ✁p2 + mc] ✁ǫ3 u4 , (7)

M2M
∗

1 = AB ū4 ✁ǫ
∗

3 [✁p1 + ✁p2 + mc] ✁ǫ2 u1 ū1 ✁ǫ3 [✁p1 − ✁p3 + mc] ✁ǫ
∗

2 u4 , (8)

where

A =
g2e

(p1 − p3)2 −m2c2
=

g2e
−2p1 · p3

, (9)

B =
g2e

(p1 + p2)2 −m2c2
=

g2e
2p1 · p2

(10)

(because p21 = m2c2, p22 = p23 = 0).

Then averaging over the initial spin states and summing over final spin states, one obtains the four

contributions

〈|M1|
2〉 =

A2

4
QµλQνκ Tr

[
γµ (✁p1 − ✁p3 +mc) γν (✁p1 +mc) γκ (✁p1 − ✁p3 +mc) γλ (✁p4 +mc)

]
, (11)

〈|M2|
2〉 =

B2

4
QµλQνκ Tr

[
γµ (✁p1 + ✁p2 +mc) γν (✁p1 +mc) γκ (✁p1 + ✁p2 +mc) γλ (✁p4 +mc)

]
, (12)

〈M1M
∗

2〉 =
AB

4
QµλQνκ Tr

[
γµ (✁p1 − ✁p3 +mc) γν (✁p1 +mc) γλ (✁p1 + ✁p2 +mc) γκ (✁p4 +mc)

]
,

(13)

〈M2M
∗

1〉 =
AB

4
QµλQνκ Tr

[
γµ (✁p1 + ✁p2 +mc) γν (✁p1 +mc) γλ (✁p1 − ✁p3 +mc) γκ (✁p4 +mc)

]
,

(14)

to

〈M2〉 = 〈|M1|
2〉+ 〈|M2|

2〉+ 〈M1M
∗

2〉+ 〈M2M
∗

1〉 . (15)

In Eqs. (11-14) the tensor Qµλ is defined

Qµλ =
∑

s=1,2

ǫ(s)µ ǫ
(s)∗
λ ,

where the photon polarization vector ǫµ = (ǫ0, ǫ) with ǫ0 = 0. The completeness relation (for the

photon polarization three vectors) is

∑

s=1,2

ǫ
(s)
i ǫ

(s)∗
j = δij − p̂ip̂j (16)

where p̂i is the unit vector p̂i = pi/|pi| (no summation). Hence

Qµλ =

{

0 , if µ = 0 or λ = 0

δµλ − p̂µp̂λ , otherwise .
(17)

Eqns. (11–17) concur with the solution manual written by D. Griffiths (available at www.academia.edu

and elsewhere).

A. Evaluation of 〈|M|2〉 in terms of Mandelstam variables, using FORM

It can be seen from Eqs. (11, 12) that 〈|M1|
2〉 and 〈|M2|

2〉 are equivalent under the substitution

p2 ↔ −p3 (or s ↔ u in terms of the Mandelstam variables s ≡ (p1 + p2)
2 and u ≡ (p2 − p4)

2 used

below), and the same symmetry links 〈M1M
∗

2〉 and 〈M2M
∗

1〉.

*** compton_scattering.frm JDW 29 Oct. 2019 ***

Vectors p1, p2, p3, p4;

Symbols m,s,u,[ge^4],[2ge^4],denu,dens;

Indices alpha,beta,gamma,delta,rho,kappa,lambda,mu,nu;

Off Statistics;

Local traceM1M1 = d_(mu,lambda)*d_(nu,kappa)

*g_(1,mu)*(g_(1,p1)-g_(1,p3)+m)

*g_(1,nu)*(g_(1,p1)+m)

*g_(1,kappa)*(g_(1,p1)-g_(1,p3)+m)

*g_(1,lambda)*(g_(1,p4)+m);

Local traceM2M2 = d_(mu,lambda)*d_(nu,kappa)

*g_(2,mu)*(g_(2,p1)+g_(2,p2)+m)

*g_(2,nu)*(g_(2,p1)+m)

*g_(2,kappa)*(g_(2,p1)+g_(2,p2)+m)

*g_(2,lambda)*(g_(2,p4)+m);

Local traceM1M2str = d_(mu,lambda)*d_(nu,kappa)

*g_(3,mu)*(g_(3,p1)-g_(3,p3)+m)

https://www.academia.edu/3851437/Griffiths-Complete_Solutions_Manual_Introduction_To_Elementary_Particles.pdf

*g_(3,nu)*(g_(3,p1)+m)

*g_(3,lambda)*(g_(3,p1)+g_(3,p2)+m)

*g_(3,kappa)*(g_(3,p4)+m);

Local traceM2M1str = d_(mu,lambda)*d_(nu,kappa)

*g_(4,mu)*(g_(4,p1)+g_(4,p2)+m)

*g_(4,nu)*(g_(4,p1)+m)

*g_(4,lambda)*(g_(4,p1)-g_(4,p3)+m)

*g_(4,kappa)*(g_(4,p4)+m);

Local M1M1 = div_(1,2)*[2ge^4]*traceM1M1*div_(1,4)*(-1/(2*p1.p3))^2;

Local M2M2 = div_(1,2)*[2ge^4]*traceM2M2*div_(1,4)*( 1/(2*p1.p2))^2;

Local M1M2str = div_(1,2)*[2ge^4]*traceM1M2str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local M2M1str = div_(1,2)*[2ge^4]*traceM2M1str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local [<|M|^2>]=M1M1+M2M2+M1M2str+M2M1str;

Trace4,1;

Trace4,2;

Trace4,3;

Trace4,4;

.sort;

repeat;

id p2.p2=0;

id p3.p3=0;

id p1.p1=m^2;

id p4.p4=m^2;

endrepeat;

Print;

.sort;

*** Introduce Mandelstam variables.

*** id s= 2*p1.p2+m^2;

*** id u=-2*p1.p3+m^2;

***

repeat;

id 1/(p1.p2)=2/(s-m^2);

id 1/(p1.p3)=-2/(u-m^2);

id p1.p4 = (u+s)/2;

id p2.p3=(u+s)/2-m^2;

id p2.p4 = (m^2-u)/2;

id p3.p4 = (s-m^2)/2;

endrepeat;

.sort;

repeat;

id s/(s-m^2)=1+m^2/(s-m^2);

id 1/(s - m^2)/(s + m^2)=1/(s^2-m^4);

endrepeat;

.sort;

id 1/(u-m^2)=denu;

id 1/(s-m^2)=dens;

.sort;

id 1/(s - m^2)/(u - m^2)*u^2=dens*denu*u^2;

.sort;

repeat;

id denu*u = 1+m^2*denu;

endrepeat;

.sort;

Bracket [2ge^4],m;

Print;

.sort;

.end;

Here are some selected outputs from the program.

traceM1M1 =

+ m^2 * ( - 16*u - 80*p1.p3 )

+ m^4 * ( 48 )

+ 16*p1.p3*s;

traceM2M2 =

+ m^2 * ( - 16*s + 80*p1.p2 )

+ m^4 * ( 48 )

- 16*p1.p2*u;

showing the expected symmetry of |M1|
2 and |M2|

2. The result for 〈|M|2〉 is (correctly)

[<|M|^2>] =

+ [2ge^4] * ( - u*dens - s*denu )

+ m^2*[2ge^4] * ( 5*dens + 5*denu )

+ m^4*[2ge^4] * ( 4*dens^2 + 8*denu*dens + 4*denu^2 );

where denu ≡ 1/(u−m2) and dens ≡ 1/(s−m2). (Thanks to Jos Vermaseren for helpful sugges-

tions.)

B. Evaluation of the differential cross-section in the Lab frame, using FORM

Defining the lab frame as that in which the electron is initially at rest and taking (ω, ω′) as respec-

tively the frequencies of the (incoming, outgoing) photons, the classical (deterministic) formula for

Compton scattering is
1

ω′
−

1

ω
=

~

mc2
(1− cos θ) , (18)

where the scattering angle θ is the angle between the outgoing and incoming photon momenta.

This can be easily rearranged to read (with ~ = c = 1)

−

(
ω′

ω

)2

sin2 θ =
m2

ω2

(

1− 2
ω′

ω
+

ω′2

ω2

)

+
2m

ω

(
ω′2

ω2
−

ω′

ω

)

, (19)

an expression that will be used below to simplify the output of a FORM program intended to

derive the Klein-Nishina formula for the differential cross section, viz.

dσ

d cos θ
=

g4e
16πm2

[
ω′3

ω3
+

ω′

ω
−

ω′2

ω2
sin2 θ

]

. (20)

In the lab. frame the four-momenta of the four particles are:

p1 = (m, 0, 0, 0) , (21)

p2 = (ω, 0, 0, ω) (22)

p3 = (ω′, ω′ sin θ, 0, ω′ cos θ) (23)

p4 = (E4L, p4x, p4y, p4z) , (24)

and it is straightforward to show that

s = m2 + 2mω , (25)

u = m2 − 2mω′ , (26)

p1.p2 = p3.p4 = mω , (27)

p1.p3 = p2.p4 = mω′ , (28)

p2.p3 = ωω′ (1− cos θ) . (29)

The differential cross section in the lab. frame is1 (Griffiths’ (1987, Eq. B.11)

dσ

dΩ
=

(
~E3

8πmcE2

)2

〈|M|2〉 →

(
1

8πm

)2 (
ω′

ω

)2

〈|M|2〉 . (30)

Writing dΩ = −d(cos θ) dφ and assuming azimuthal symmetry for the integration 0 ≤ φ ≤ 2π we

obtain
dσ

d cos θ
= −2π

dσ

dΩ
. (31)

Below, FORM is used to evaluate Eq. (31), checking against Eq. (20).

*** compton_scattering_crossection_lab.frm JDW 16 Nov. 2019 ***

*** Aim: reproduce the Klein-Nishina formula ***

Vectors p1, p2, p3, p4;

Symbols [2Pi],[2ge^4],[(hbar/8mPi)^2],[ge^4/16Pim^2],[(op/o)^2 sin^2 theta];

Symbols m,omega,omegaprime,oONop,opONo,denomOmega,mopONo2,m2opONo3,m2overo2,m2op2ONo2;

Indices alpha,beta,gamma,delta,rho,kappa,lambda,mu,nu;

Off Statistics;

Local traceM1M1 = d_(mu,lambda)*d_(nu,kappa)

*g_(1,mu)*(g_(1,p1)-g_(1,p3)+m)

1This is presumably given, too, in the 2nd edition – which however is currently inaccessible to the writer.

*g_(1,nu)*(g_(1,p1)+m)

*g_(1,kappa)*(g_(1,p1)-g_(1,p3)+m)

*g_(1,lambda)*(g_(1,p4)+m);

Local traceM2M2 = d_(mu,lambda)*d_(nu,kappa)

*g_(2,mu)*(g_(2,p1)+g_(2,p2)+m)

*g_(2,nu)*(g_(2,p1)+m)

*g_(2,kappa)*(g_(2,p1)+g_(2,p2)+m)

*g_(2,lambda)*(g_(2,p4)+m);

Local traceM1M2str = d_(mu,lambda)*d_(nu,kappa)

*g_(3,mu)*(g_(3,p1)-g_(3,p3)+m)

*g_(3,nu)*(g_(3,p1)+m)

*g_(3,lambda)*(g_(3,p1)+g_(3,p2)+m)

*g_(3,kappa)*(g_(3,p4)+m);

Local traceM2M1str = d_(mu,lambda)*d_(nu,kappa)

*g_(4,mu)*(g_(4,p1)+g_(4,p2)+m)

*g_(4,nu)*(g_(4,p1)+m)

*g_(4,lambda)*(g_(4,p1)-g_(4,p3)+m)

*g_(4,kappa)*(g_(4,p4)+m);

Local M1M1 = div_(1,2)*[2ge^4]*traceM1M1*div_(1,4)*(-1/(2*p1.p3))^2;

Local M2M2 = div_(1,2)*[2ge^4]*traceM2M2*div_(1,4)*( 1/(2*p1.p2))^2;

Local M1M2str = div_(1,2)*[2ge^4]*traceM1M2str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local M2M1str = div_(1,2)*[2ge^4]*traceM2M1str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local [<|M|^2>]=M1M1+M2M2+M1M2str+M2M1str;

Local dsigmadomegaLAB = [(hbar/8mPi)^2]*[<|M|^2>]*omegaprime^2*1/omega^2;

Local dsigmadcosthetaLAB = [2Pi]*dsigmadomegaLAB;

Trace4,1;

Trace4,2;

Trace4,3;

Trace4,4;

.sort;

repeat;

id p2.p2=0;

id p3.p3=0;

id p1.p1=m^2;

id p4.p4=m^2;

endrepeat;

.sort;

*** No need to introduce Mandelstam variables, can evaluate the dot products

*** directly in terms of omega, omega’ ***

repeat;

id 1/(p1.p2) = 1/(m*omega);

id 1/(p1.p3) = 1/(m*omegaprime);

id p1.p4 = m^2 + m*(omega-omegaprime);

id p2.p3 = m*(omega-omegaprime);

id p2.p4= m*omegaprime;

id p3.p4 = m*omega;

endrepeat;

.sort;

repeat;

*id omega/omegaprime=oONop;

*id omegaprime/omega=opONo;

endrepeat;

Print;

.sort;

id 1/omega=denomOmega;

id denomOmega*omegaprime=opONo;

.sort;

id [2Pi]*[2ge^4]*[(hbar/8mPi)^2]=[ge^4/16Pim^2];

.sort;

Bracket [ge^4/16Pim^2],m,omega;

Print;

.sort;

*** at this point the correct answer is recognisable

*** need now to simplify, pulling in the sin^2(theta) term

id denomOmega*opONo*m=mopONo2;

id denomOmega*mopONo2*m=m2opONo3;

id denomOmega*denomOmega*m*m=m2overo2;

id mopONo2*mopONo2=m2op2ONo2;

.sort;

*** The following four steps amount to a "fudge", to recast the solution, which is recognizably

*** into the standard Klein-Nishina expression involving sin^2(theta)

if(coefficient<0) discard;

if (coefficient > 1) Discard;

id m2op2ONo2=0;

id m2overo2=-[(op/o)^2 sin^2 theta];

.sort;

Bracket [ge^4/16Pim^2],m,omega;

Print;

.sort;

.end;

An intermediate output is

dsigmadcosthetaLAB =

+ [ge^4/16Pim^2] * ( opONo + opONo^3 )

+ [ge^4/16Pim^2]*m * ( - 2*opONo*denomOmega + 2*opONo^2*denomOmega )

+ [ge^4/16Pim^2]*m^2 * ( denomOmega^2 - 2*opONo*denomOmega^2 + opONo^2*

denomOmega^2 );

where for instance opONo represents ω′/ω. The expected term in sin2 θ is recognizable, under

disguise in the second two lines. How to make the needed transformation? After an operation that

amounts (one could say) to the electronic equivalent of using the old typewriter white-out, and

that presumably can be achieved elegantly by the experienced user of FORM), one obtains

dsigmadcosthetaLAB = [ge^4/16Pim^2] * ( opONo + opONo^3 - [(op/o)^2 sin^2 theta] );

This concurs with the Klein-Nishina formula.

